
                                             Sortierverfahren      

Für das Sortierverfahren TAUSCH-SORT gelten diese Überlegungen:

1) Wir legen Spielkarten auf den Tisch und gehen die Reihe durch.

2) Ist die erste Karte größer als die zweite, so wird getauscht. Dann vergleichen wir 

    die zweite mit der dritten Karte, usw. Das wird fortgesetzt bis zum Ende der Reihe.

3) Die Karten sind jetzt noch nicht vollständig sortiert, also beginnen wir wieder bei 

    der ersten Karte und gehen die Reihe durch.

4) Wir zählen wie oft Karten verglichen werden und wie oft Karten vertauscht werden.

Grundlagen für die Sortierverfahren

1) Buchstaben und Zeichen haben einen ANSI-Wert:  A= 65 , B= 66, damit gibt es größer und kleiner.

2) Die Daten sind vom Typ array,  der Zugriff erfolgt über einen Index

var zuf: array [0..100] of integer;

3) Anwenden der Vergleichsoperation:    wenn  a[i] > a[i+1] dann.

4) Zählen wie oft ein Vergleich und eine Vertauschung der Zahlen stattfindet.

Der Algorithmus

1) Gehe die Liste durch      (1. Schleife )

2) if  a[i]  >  a [i+1] dann tausche , fahre fort bis zum Ende der Reihe.   

3) Beginne wieder bei 1 .    (2. Schleife )

Der Vorgang  <TAUSCHEN> hat 3 Teilschritte:  

if  a[i]  >  a [i+1] then begin   

         hilf := a [i];

         a [i]:= a [i+1];

         a [i+1] := hilf;  

         inc (tausch);

end;
Die Algorithmen für die Sortierverfahren
Tausch-sort

1) Zwei Schleifen. Die 1. Schleife geht von i= 0 bis max, 

     die 2. Schleife geht von j= 0 bis max -1.         // das letzte Element nicht !

2) Man geht jede Schleife durch. Sind 2 Elemente nicht richtig, werden sie getauscht.

Bubble-sort

1) Zwei Schleifen. Die 1. Schleife geht von i= 0 bis max, 

     die 2. Schleife geht abwärts von j= max bis i. (Dadurch wird die Anzahl der  

     Vergleiche bei jedem Durchgang der 2. Schleife kleiner).

2) Sind 2 Elemente nicht richtig, werden sie getauscht.

Auswählen-sort

1)  Zwei Schleifen. Die 1. Schleife geht von i = 0 bis max, 

     die 2. Schleife geht von j = i bis max. 

2)  Nimm das Element  a [i] der Liste    (1. Schleife)

3)  suche das kleinste Element a [j] in der Rest-Menge N,    (2. Schleife)

4)  tausche es gegen das Element a [i],

5)  erhöhe i und beginne wieder bei 2.
Einfügen-sort

1)  Zwei Schleifen. Die 1. Schleife geht von i = 1 bis max, 

     die 2. Schleife geht von j = i bis max. 

2) Wir teilen die Liste in einen linken Teil (= sortierte Liste) und einen rechten Teil

     (= unsortierten Liste). Der linke Teil besteht anfangs aus einem Element.

3) Nimm das Element a [j] der unsortierten Liste,

4) füge es im linken Teil, und an der richtigen Stelle, der sortierten Liste ein,

5) erhöhe j und beginne wieder bei 3 .

Quick-sort 

Sortieren durch zerlegen ( C. A. Hoare 1955

1. Der linke und der rechte Rand werden mit i und j markiert.

2. Man nimmt das mittlere Element und teilt das Feld in 2 Teile.

    mitte := (i +j) div 2;   71 = a [mitte] ;

3. Man geht das linke Feld durch, inc (i) und sucht eine Zahl grösser der mittleren,

    man geht das rechte Feld durch, dec (j) und sucht eine Zahl kleiner der mittleren.

4. Man tauscht beide Zahlen.

5. Fahre fort bei 2, bis i > j ist.

6. Die beiden Teilmengen werden wieder geteilt und wie unter 2)   behandelt.

7. Die Unterteilung ist beendet, wenn die Teilmengen aus einem oder zwei Elementen bestehen. 

8.  Der Algorithmus ist rekursiv.

Die einzelnen Sortierverfahren
1. TAUSCH_ Sort

2. BUBBLE_Sort

3. Auswählen_Sort (straight selection)

4. Einfügen_Sort (straight insertion)

5. QUICK_Sort

6. SHAKER_Sort

7. SHELL_Sort

Wir schreiben die Prozeduren und speichern sie in der Unit  "UNIT_SORTIERVERFAHREN"  ab.

unit Unit_sortierverfahren;

interface

procedure TAUSCH_SORT (var feld: array of integer; max: integer;

       var vergleich, tausch: integer);

procedure BUBBLE_SORT (var feld: array of integer; max: integer;

       var vergleich, tausch: integer);

procedure AUSWAHLEN_SORT (var feld: array of integer; max: integer;

       var vergleich, tausch: integer);

procedure EINFUGEN_SORT (var feld: array of integer; max: integer;

       var vergleich, tausch: integer); 

procedure QUICK_SORT (var feld: array of integer; links, rechts: integer; 

       var vergleich, tausch: integer);

implementation

// -------------------------

procedure TAUSCH_SORT (var feld: array of integer; max: integer; var vergleich, tausch: integer);

var hilf, i, j : integer;

begin

     for i:= 0 to max do begin

         for j:= 0 to max-1 do begin          // das letzte Element wird nicht bearbeitet

             inc (vergleich);

             if feld[j] > feld [j+1] then begin

                  hilf := feld [j];  feld[j]:= feld[j+1];  feld[j+1] := hilf;

                  inc (tausch);

             end;         end;       end;      end;

//----------------------------

procedure BUBBLE_SORT (var feld: array of integer; max: integer; var vergleich, tausch: integer);

var i, j, hilf: integer;

begin

     for i:= 1 to max do begin

         for j:= max downto i do begin

             inc(vergleich);

             if feld[j] < feld [j-1] then begin

                  hilf:= feld[j-1]; feld[j-1]:= feld[j]; feld[j]:= hilf;

                  inc(tausch);

             end;         end;       end;      end;

//-----------------------------

procedure AUSWAHLEN_sort (var feld: array of integer; max: integer; var vergleich, tausch: integer);

var i, j, hilf, min : integer;

begin

     for i:= 0 to max do begin

         hilf:= feld[i];

         min:= i;

         for j:= i+1 to max do begin

             inc(vergleich);

             if feld[j] < hilf then begin

                  hilf:= feld[j];

                  min:= j;

             end;

         end;

         hilf:= feld[i]; feld[i]:= feld[min]; feld[min]:= hilf;

         inc(tausch);

     end;

end;

//----------------------------

Procedure EINFUGEN_sort (var feld: array of integer; max: integer; var vergleich, tausch: integer);

var i, j, hilf: integer;

begin

     for i:= 1 to max do begin

         hilf:= feld[i];

         j:= i;

         while (hilf < feld[j-1]) and (j > 0) do begin

               feld[j]:= feld[j-1];

               j:= j-1;

               inc(vergleich);

         end;

         feld[j] := hilf;

         inc(tausch);

     end;

end;

//-----------------------------------

procedure QUICK_sort (var feld: array of integer; links, rechts: integer; var vergleich, tausch: integer);

var i, j, hilf, mitte: integer;

    begin

         i:= links; j:= rechts;

         mitte:= feld[(links + rechts) div 2];      // mittleres element

         repeat

               while feld[i] < mitte do begin

                     i:= i+1;

                     inc(vergleich);

               end;

               while feld[j] > mitte do begin

                     j:= j-1;                      // zur Mitte hin annähern

                     inc(vergleich);

               end;

               if i <=j then begin

                    hilf := feld[i];

                    feld[i] :=feld[j]; 

                    feld[j]:= hilf;

                    i := i+1;

                    j := j-1;

                    inc(tausch);

               end;

         until i > j;

         if links < j then quick_sort (feld, links, j, vergleich, tausch);

         if rechts > i then quick_sort (feld, i, rechts, vergleich, tausch);

end;

//-----------------------------

procedure SHAKER_sort(var feld: array of integer; max: integer; var vergleich, tausch: integer);

 var k,j,links, rechts, x:integer;

 begin

      links:=low(feld)+1;  rechts:= max;

      repeat

            for j:= max downto links do begin

                inc(vergleich);

                if feld[j-1]>feld[j] then begin

                     inc(tausch);

                     x:=feld[j-1];

                     feld[j-1]:=feld[j];

                     feld[j]:=x;

                     k:=j;

                end;

             end;

             links:=k+1;

             for j:=links to rechts do  begin

                inc(vergleich);

                if feld[j-1]>feld[j] then begin

                     inc(tausch);

                     x:=feld[j-1];

                     feld[j-1]:=feld[j];

                     feld[j]:=x;

                     k:=j;

                end;

             end;

             rechts:=k-1;

      until links > rechts;

end;

 //---------------------

procedure SHELL_sort (var feld: array of integer; max: integer; var vergleich, tausch: integer);

var i, j, k, hilf: integer;                      // aus Buch Doberenz S. 871

begin

    k:= max div 2;

    while k > 0 do begin

        for i:= 0 to max- k do begin

            j:= i;

            while (j >= 0) and (feld[j] > feld[j+ k]) do begin

                hilf:= feld[j];

                feld[j]:= feld[j+ k];

                feld[j+ k] := hilf;

                dec(j,k);

                inc(tausch);

            end;

            inc(vergleich);

        end;

        k := k div 2;

    end;

end;

Aufgabe

Wir erzeugen 100 Zufallszahlen und speichern sie in der Datei "Liste1.DAT"

procedure TForm1.Button1Click(Sender: TObject);     // die Liste der Zahlen erstellen

var i, xx: integer;

begin

     listBox1.clear;

     randomize;

     for i:= 0 to 99 do begin               //  mit der Null sind es 100 Zahlen

          xx:= random (100);

          listBox1.Items.add (intTostr (xx));

     end;

     listBox1.items.saveToFile ('Liste1.DAT');

end;

end.

Form erstellen

/ listBox

/ radioGroup  / Items = Name der Sortierverfahren

Aufgabe

1)  Die Zahlen werden aus der Datei  "Liste1.DAT" gelesen, in das array zuf [i] 
     gebracht,  und in der ListBox1 ausgegeben.

2)  Die Zahlen werden mit einem Verfahren sortiert und in der ListBox2 ausgegeben.

3)  Wir füllen die Tabelle aus.

	
	Anzahl von Vergleich
	Formel
	Anzahl von Tausch
	Formel

	Tausch -sort
	
	
	
	

	Bubble -sort
	
	
	
	

	Auswählen -sort
	
	
	
	

	Einfügen -sort
	
	
	
	

	Quick -sort
	
	
	
	


Methoden

1) Procedure mit Wert- Übergabe

     procedure TAUSCH_SORT (var feld: array of integer; max: integer; 

                     var vergleich, tausch: integer);

2) uses -Methode

    Die Sortierverfahren sind in der Unit  "Unit_Sortierverfahren"  enthalten. 

const max = 99;                              //  die Feldgröße

var  zuf: array [0..max] of integer;   // der array muß mit NULL  beginnen, es gibt sonst Probleme

{$R *.DFM}

uses Unit_Sortierverfahren

procedure TForm1.Button1Click(Sender: TObject);     // Zahlen ausgeben

var i: integer;

begin

     listBox1.clear;

     listBox1.items.LoadFromFile ('Liste.DAT');

     for i:= 0 to listBox1.items.count-1 do

         zuf[i] := strToint (listBox1.Items[i]);           // und in array  zuf [i]  einlesen

end;

procedure TForm1.Button2Click(Sender: TObject);        // Zahlen sortieren

var i: integer;

     vergleich, tausch: integer;

begin

     vergleich:= 0;

     tausch:= 0;

     case radioGroup1.ItemIndex of

         0: TAUSCH_sort (zuf, max, vergleich, tausch);

         1: BUBBLE_sort (zuf, max, vergleich, tausch);

         2: AUSWAHLEN_sort (zuf, max, vergleich, tausch);

         3: EINFUGEN_sort (zuf, max, vergleich, tausch);

         4: QUICK_sort (zuf, 0, vergleich, tausch);         // links =0, rechts = max

     end;

     for i:= 0 to max do  listBox2.Items.add (intToStr (zuf[i]));

     edit1.text:= intTostr (vergleich);

     edit2.text:= intTostr (tausch);

end;

Aufwand der Sortierverfahren

	
	Vergleich
	Formel
	Tausch
	Formel

	Tausch-Sort
	9900
	N2 -N
	2480
	1/4  N2

	Bubble-Sort
	5000
	1/2  N2
	2500
	1/4  N2

	Auswählen-Sort
	4950
	1/2  N2
	100
	 N

	Einfügen-Sort
	2500
	1/4  N2
	100
	 N

	Quick-Sort
	350
	 4 * N 
	200
	N * log (N)


Der Aufwand eines Sortier-Algorithmus wird bestimmt durch

- die Sortierzeit für die Zahlen, d.h. die Anzahl von Tausch und Vergleich,

   ( die Sortierzeit wird kleiner, wenn die Zahlen über größere Entfernungen hin 

   vertauscht werden),

- die Zeit zum Programmieren,

- der Bedarf an Speicherplatz, d.h. ob das Verfahren iterativ oder rekursiv ist wie 

   bei Quicksort.

- wie die Zahlenliste bereits vorliegt.

   ( unsortiert, aufsteigen, absteigend, Fehler am Anfang, ...)

Aufgabe             (Eine Erweiterung von Programm  sort 3 : Sortierverfahren).

1) Es werden verschiedene Zahlenlisten erstellt und abgespeichert.
2) Verschiedene Sortierverfahren werden auf diese Zahlenlisten angewendet.

3) Die Werte für Vergleich und Tausch werden in einem StringGrid ausgegeben. 

Form erstellen

/listBox

/stringGrid

Methoden

{$R *.DFM}

uses Unit_Sortierverfahren;

procedure TForm1.FormCreate(Sender: TObject);

begin

     stringgrid1.Cells[0,0]:='Verfahren';

     stringgrid1.Cells[1,0]:='Vergleich';

     stringgrid1.Cells[2,0]:='Tausch';

end;

procedure TForm1.Button2Click(Sender: TObject);  // listen auswählen und Zahlen ausgeben

var i: integer;

begin

     listbox1.clear;

     listbox2.clear;

     case radiogroup1.itemindex of

          0:listBox1.items.loadFromFile ('Liste1.dat');      // unsortiert

          1:listBox1.items.loadFromFile ('Liste2.dat');      // Fehler an Anfang

          2:listBox1.items.loadFromFile ('Liste3.dat');      // Fehler am Ende

          3:listBox1.items.loadFromFile ('Liste4.dat');      // absteigend sortiert

     end;

end;

procedure TForm1.Button1Click(Sender: TObject);   //  die Zahlen sortieren 

var i: integer;

      vergleich, tausch: integer;

begin

     listBox2.clear;

     for i:= 0 to listBox1.items.count -1 do     // die zahlen in den array einlesen

         zuf[i]:= strToint (listbox1.items[i]);

     case radiogroup2.itemindex of

          0: Tausch_Sort(zuf, max, vergleich, tausch);

          1: Bubble_Sort(zuf, max, vergleich, tausch);

          2: Auswahlen_Sort(zuf, max, vergleich, tausch);

          3: Einfugen_Sort (zuf, max, vergleich, tausch);

          4: Quick_Sort (zuf, 0, max, vergleich, tausch);

     for i:= 0 to max do

     listbox2.items.add (inttostr (zuf[i]));            

     inc (zeile);

     stringgrid1.cells[0, zeile]:=radiogroup2.items[radiogroup2.itemindex]+'   '+ radiogroup1.items [radiogroup1.itemindex];

     stringgrid1.cells[1, zeile]:=inttostr (vergleich);

     stringgrid1.cells[2, zeile]:=inttostr (tausch);

end;

Aufgabe

1. Ein Sortier-Verfahren auswählen.

2. Nach <ButtonClick> werden die Panel automatisch sortiert. Das ausgewählte Sortierverfahren wird durch die Bewegung der Panel veranschaulicht.

Form erstellen

/  RadioGroup,  / columns = 5

/ 10 große Panel beschriftet mit willkürlichen Zahlen,

/ 2 kleine farbige Panel, die als Zeiger für die zwei Schleifen dienen.

/  label,  /  visible = false

const max = 9;       // mit der NULL sind es 10 Zahlen

var

   Form1: TForm1;

   brett: array [0..max] of TPanel;

   ttext: array [0..max] of string;

   min_grue, min_rot, max_grue: integer;

implementation

 {$R *.DFM}

procedure TForm1.FormCreate(Sender: TObject);

var i: byte;

begin

     min_rot:= panel12.left;                  // roter Zeiger

     min_grue:= panel13.left;                 // grüner Zeiger

     for i:= 0 to max do begin

          brett [i] := tPanel (Components [i]);

          ttext [i]:= brett [i].caption;      // die Beschriftung der Panel wird gespeichert

          max_grue:= brett[i].left +19;

     end;

end;

procedure TForm1.Button2Click(Sender: TObject);     // Sort-verfahren auswählen

var i: byte;

begin

     panel12.left:= min_rot;                  // die farb-panel zurücksetzen

     panel13.left:= min_grue;

     for i:= 0 to max do begin

         brett [i].color:= clBtnFace;

         brett [i].caption:= ttext [i];       // alte Beschriftung wieder herstellen

     end;

     case radioGroup1.Itemindex of

          0: TAUSCH_SORT;

          1: BUBBLE_SORT;

          2: AUSWAHLEN_SORT;

          3: EINFUEGEN_SORT;

          4: QUICK_SORT (0, max);

     end;

end;

procedure TForm1.TAUSCH_SORT ;

var i, j: integer;

begin

     for i:=0 to max do begin

         panel13.Left:= min_grue;                   // grünes panel an den anfang bringen

         Pause (1000);

         Panel12.left:= brett[i].left +19;           // rotes panel -- 1.Schleife

         for j:=0 to max-1 do begin

             pause(500);

             Panel13.left:= brett[j+1].left +19;     // grünes panel -- 2.Schleife

             if brett[j].caption > brett [j+1].caption then

                    TAUSCHEN_1 (j, j+1);

         end;

     end;

end;

procedure tform1.TAUSCHEN_1 (x, y: byte);  // die panel tauschen

var hilf: integer;

    hilf_brett: TPanel;

begin

      pause (500);

      brett[y].top:= brett[y].top +70;     // panel geht nach unten

      pause(500);

      hilf:= brett[y].left;

      brett[y].left:= brett[x].left;       // X-panel geht nach links

      pause(500);

      brett[x].left:= hilf;                // Y-panel geht nach rechts

      pause (500);

      brett[y].top:= brett[y].top -70;     // panel geht nach oben

      pause (500);

      hilf_brett:= brett [x];              // die panel Index Nummern tauschen

      brett[x]:= brett[y];

      brett[y]:= hilf_brett;

end;

procedure Tform1.BUBBLE_SORT;

var i, j: integer;

begin

     for i:=0 to max-1 do begin

          panel13.left:= max_grue;                  // grünes panel an das Ende bringen

          pause(1000);

          panel12.left:= brett[i].left +19;          // das rote panel

          for j:= max-1 downto i do begin

              panel13.left:= brett[j+1].left +19;     // das grüne panel

              pause (500);

              if brett [j].caption > brett [j+1].caption then

                    TAUSCHEN_1 (j+1, j);

          end;

     end;

end;

procedure TForm1.AUSWAHLEN_SORT;

var i, j, k, merke, min: integer;

    hilf: string;

    hilf_brett: tPanel;

begin

     for i:= 0 to max-1 do begin

         Panel12.left:= brett[i].left +19;            // rotes panel -- 1.Schleife

         pause (1000);

         Panel13.left:= brett[i].left +19;            // das grüne panel

         pause (500);

         hilf:= brett[i].caption;

         for j:= i to max do begin                    // Minimum suchen

             Panel13.left:= brett[j].left +19;        // das grüne panel

             Pause (500);

             if hilf >= brett[j].caption then begin

                   hilf:= brett[j].caption;

                   min:= j;

                   k:= j;

             end;

         end;

//-------------

         brett[min].color:= cllime;                 // Minimum anfärben

         merke:= brett[min].left;

         brett[min].top:= brett[min].top +70;       // panel [min] geht nach unten

         pause(500);

         while k > i do begin                       // panel [min] geht nach links

               brett[min].left:= brett[k-1].left;

               pause (500);

               k:= k-1;

         end;

         brett[i].left:= merke;                     // geht [i] nach rechts

         pause (500);

         brett[min].top:= brett[min].top -70;       // panel [min] geht nach oben

         pause(500);

         hilf_brett:= brett[i];                     // die panel Index Nummern tauschen

         brett[i]:= brett[min];

         brett[min]:= hilf_brett;

         pause (500);

     end;

end;

Procedure TForm1.EINFUEGEN_sort;

var i, j, left: integer;

    hilf_brett: tPanel;

begin

     for i:= 0 to max do begin

         Panel12.left:= brett[i].left +19;          // rotes panel -- äußere Schleife

         pause (1000);

         Panel13.left:= brett[i].left +19;          // grünes panel -- die Verschiebe-Stelle

         pause (500);

         j:= i;

         brett[j].top:= brett[j].top +70;           // panel nach unten

         while (brett[j].caption < brett[j-1].caption) and (j > 0) do begin

              Panel13.left:= brett[j-1].left +19;   // das grüne panel geht nach links

              pause(500);

              left:= brett[j].left;

              brett[j].left:= brett[j-1].left;      // geht nach rechts

              pause (500);

              brett[j-1].left:= left;

              hilf_brett:= brett[j];                // die panel Index Nummern tauschen

              brett[j]:= brett[j-1];

              brett[j-1]:= hilf_brett;

              pause (500);

              j:= j-1;

         end;

         brett[j].top:= brett[j].top -70;

         brett[j].color:= clLime;

      end;

end;

procedure TForm1.Quick_Sort (links, rechts: integer);

var i, j, mitte: integer;

begin

     panel12.left:= brett[links].left +19;           // rotes panel nach links

     panel13.left:= brett[rechts].left +19;          // grünes panel nach rechts

     i:= links;

     j:= rechts;

     mitte:= (links + rechts) div 2;

     repeat                                        

           brett [mitte].color:= clAqua;

           pause (1000);                           

           while               // es wird der kleinste wert in der linken hälfte gesucht

                 brett[i].caption < brett[mitte].caption do begin

                 panel12.left:= brett[i].left +19;   // rotes panel

                 pause (2000);

                 i:= i+1;

           end;

           while               // es wird der größte wert in der rechten hälfte gesucht

                 brett[j].caption > brett[mitte].caption do begin

                 panel13.left:= brett[j].left +19;   // grünes panel

                 pause (2000);

                 j:= j-1;

           end;

           if i <= j then begin

               TAUSCHEN_2 (i, j);

               i:=i +1;

               j:=j -1;

           end;

     until i > j;

     brett [mitte].color:= clBtnface;            // die Mitte wieder normal färben

     if links < j then Quick_Sort (links, j);

     if rechts > i then Quick_Sort (i, rechts);

end;

procedure tForm1.TAUSCHEN_2 (x, y: byte);        // die panel tauschen

var hilf_brett: tPanel;

    a, b: integer;

begin

     a:= brett[x].left;

     b:= brett[y].left;

     brett[x].top:= brett[x].top +70;            // panel nach unten

     brett[y].top:= brett[y].top -70;

     pause (500);

     while brett[x].left < b-50 do begin         // die panel verschieben

           brett[x].left := brett[x].left +58;

           pause(300);

     end;

     while brett[y].left > a+50 do begin

           brett[y].left := brett[y].left -58;

           pause (300);

     end;

     brett[x].top:= brett[x].top -70;

     brett[y].top:= brett[y].top +70;            // panel nach oben

     hilf_brett:= brett[x];                            // die panel tauschen

     brett[x]:= brett[y];

     brett[y]:= hilf_brett;

     pause (1000);

end;

END.

Sortieralgorithmen

Die Problemstellung ist einfach und klar definiert: Gegeben ist ein Array mit Elementen, die nach ihrem Schlüssel sortiert werden können, über deren momentane Anordnung jedoch nichts bekannt ist (das heißt natürlich auch, dass sie theoretisch schon sortiert sein könnten oder genau in der umgekehrten Reihenfolge vorliegen könnten). Der Sortieralgorithmus soll die Elemente in die richtige Reihenfolge bringen.


Bubblesort

Bubblesort ist einer der simpelsten Sortieralgorithmen. Die Idee dahinter ist die folgende: Im ersten Durchgang wird das Array vom Ende bis zum Anfang durchgegangen und in jedem Schritt das aktuelle Element mit dem nächsten verglichen. Ist das hintere kleiner, werden die beiden vertauscht, so dass größere Elemente am Weg liegen bleiben und das jeweils kleinste weiter nach vorn wandert. Am Anfang angekommen, liegt das kleinste Element des Arrays an Position 1, wo es auch hin soll.
Im nächsten Durchlauf wird dann das zweitkleinste Element gesucht und nach vorne durchgereicht. Logischerweise muss dieser Durchlauf nicht ganz bis zum Anfang gehen, sondern nur bis Index 2. Im dritten Durchlauf braucht nur bis Index 3 gegangen zu werden usw., bis das Verfahren abgeschlossen ist.

Man kann sich vorstellen, dass die kleineren Elemente "leichter" sind und daher aufsteigen wie Blasen - daher der Name Bubblesort. 

const TableSize = 20; //Array-Obergrenze
type TableArray: array[1..TableSize] of Byte;

procedure BubSort(var Table: TableArray);

var
  Run: LongInt;

  Index: LongInt;

  x: Byte;

begin
  for Run := 2 to TableSize - 1 do
    for Index := TableSize downto Run do
      if Table[Index] < Table[Index - 1] then
      begin
        x := Table[Index];

        Table[Index] := Table[Index - 1];

        Table[Index - 1] := x;

      end;

end;

Deutlich zu erkennen ist der Vertauschungsvorgang, bei dem das "untere" Element zunächst in der Variablen x zwischengespeichert wird, dann den Wert des "oberen" zugewiesen bekommt und anschließend x im "oberen" Index gespeichert wird.
Bubblesort genießt keinen besonders guten Ruf, weil es sehr langsam ist. Trotzdem gehört es zu denjenigen Algorithmen, die man häufiger antrifft, nicht zuletzt wegen seines hohen Bekanntheitsgrades. Hat man nur kleine Arrays zu sortieren (z. B. 20 bis 50 Elemente), schlägt es sich in der Regel auch ganz wacker; für größere Arrays hingegen ist es nicht zu empfehlen. Ich habe mir vorgenommen, auf dieser Seite vollkommen auf Komplexitätsanalysen zu verzichten, wenn es jemanden interessiert - die Literatur quillt über davon :-)

Selectionsort

Selectionsort ist ebenfalls ein sehr einfacher Algorithmus. Das Prinzip hinter dem "Sortieren durch Auswählen" besteht darin, dass das Array von vorn nach hinten durchgegangen wird, und für jeden Platz das passende Element herausgesucht wird.
Als erstes betrachtet man also das Element am Index 1. Man geht dann das Array Element für Element durch und merkt sich, welches das kleinste ist. Dieses vertauscht man dann mit dem ersten, so dass ganz vorne das kleinste Element steht. Im nächsten Durchgang betrachtet man den Index 2, durchsucht wiederum den Array nach dem kleinsten Element, wobei natürlich der Index 1 außen vor bleibt, und setzt das kleinste an den Index 2. Im 3. Durchgang bleiben dann die ersten beiden Elemente außen vor usw. So geht es immer weiter, bis man schließlich das vorletzte Element korrekt besetzt hat. Aus naheliegenden Gründen stimmt das letzte Element dann schon...

procedure SelSort(var Table: TableArray);

var
   Elem     : LongInt;

   Index    : LongInt;

   Smallest : Byte;

   SmallInd : LongInt;

begin
   for Elem := 1 to TableSize - 1 do
   begin
      SmallInd := Elem;

      Smallest := Table[SmallInd];

      for Index := Elem + 1 to TableSize do
      begin
         if Table[Index] < Smallest then
         begin
            Smallest := Table[Index];

            SmallInd := Index;

         end;

      end;

      if SmallInd <> Elem then
      begin
         Table[SmallInd] := Table[Elem];

         Table[Elem] := Smallest;

      end;

   end;

end;

In der Variablen Smallest merkt sich der Algorithmus die Größe des bislang kleinsten Elementes, in SmallInd seine Position. Beide werden zunächst auf Elem initialisiert, d. h. auf die Position, die besetzt werden soll. Dann wird der übrige Array durchsucht, und wenn ein kleineres Element auftaucht, werden die beiden aktualisiert.
Zwei Besonderheiten sind beim Vertauschungsvorgang zu betrachten: Zum einen wird er nur ausgeführt, wenn tatsächlich eine neue Position gefunden wurde; zum anderen benötigen wir keinen Zwischenspeicher, weil der Wert des neuen Elementes ja noch in Smallest steht.

Selectionsort ist schneller als Bubblesort, - im Allgemeinen kann man vom Faktor 2 ausgehen - für größere Arrays jedoch auch nicht besonders gut geeignet.

Insertionsort

Insertionsort steht für "Sortieren durch Einfügen". Die Idee, die dem zugrunde liegt, ist einfach: Ein Teil am Anfang des Arrays, so wird angenommen, ist schon sortiert (zu Beginn ist dieser Teil natürlich 0 Elemente groß!). Nun wird das erste Element aus dem unsortierten Teil genommen und an der richtigen Stelle in den sortierten Teil eingefügt. Der sortierte Teil wächst dabei natürlich um ein Element, bleibt aber sortiert, wohingegen der unsortierte Teil um ein Element schrumpft.
Unser Algorithmus fängt also beim ersten Element an und geht dann bis zum letzten durch. Für jedes Element macht er nun folgendes: Von seiner Position aus bewegt er sich zum Anfang hin, solange die Elemente - wir sind ja im sortierten Teil - noch größer oder gleich dem in Frage stehenden Element sind. Dabei schiebt er jedes Element, das er überquert, eins nach hinten, so dass er immer eine Lücke vor sich herschiebt. Findet er dann einen kleineren Kandidaten, schiebt er die Lücke nicht weiter, sondern setzt sein Element hinein. Auf diese Weise verbindet der Algorithmus die Suche nach der richtigen Position mit dem Verschieben der darüberliegenden Elemente. 

procedure InsSortLinear(var Table: TableArray);

var
   Elem  : LongInt;

   x     : Byte;

   Index : LongInt;

begin
   for Elem := 2 to TableSize do
   begin
      x := Table[Elem];

      Index := Elem;

      while (Index > 1) and (Table[Index - 1] >= x) do
      begin
         Table[Index] := Table[Index - 1];

         Dec(Index);

      end;

      Table[Index] := x;

   end;

end;

Wie man sieht, wird das Element zunächst in x gespeichert. Dann beginnt das Verschieben, bis entweder ein Element gefunden wird, das kleiner als x ist, oder bis man bei Index 1 angelangt ist. Im letzteren Fall hat man offenbar kein kleineres Element im sortierten Teil und plaziert das Element an Position 1.
Um die richtige Position zu finden, verwendet diese Version von Insertionsort eine lineare Suche. Da liegt die Idee nicht fern, das Verfahren dadurch zu beschleunigen, dass man eine binäre Suche verwendet. Das Verschieben der "dahinterliegenden" Elemente muss dann gesondert erledigt werden. 

procedure InsSortBinary(var Table: TableArray);

var
   Elem   : LongInt;

   L, R   : LongInt;

   Middle : LongInt;

   Index  : LongInt;

   x      : Byte;

begin
   for Elem := 2 to TableSize do
   begin
      x := Table[Elem];

      L := 1;

      R := Elem;

      while L < R do
      begin
         Middle := (L + R) div 2;

         if Table[Middle] > x then R := Middle

         else if Table[Middle] < x then L := Middle + 1

         else R := L;

      end;

      for Index := Elem downto L + 1 do Table[Index] := Table[Index - 1];

      Table[L] := x;

   end;

end;

Auch in dieser verbesserten Version ist Insertionsort leider nicht unbedingt der Überflieger. Zwar ist es das schnellste der "langsamen" Verfahren, bleibt aber hinter den schnelleren Algorithmen deutlich zurück.

Shellsort

Für die einen stellt Shellsort den idealen Kompromiss aus Einfachheit und Geschwindigkeit dar, für die anderen ist es uninteressant, weil schnellere Verfahren längst bekannt sind. Der einzige nach seinem Erfinder benannte Sortieralgorithmus hat jedenfalls nach wie vor eine große Anhängerschaft.
Die Idee ist folgende: Der Array wird zunächst in mehrere Gruppen aufgeteilt, die dadurch festgelegt werden, daß zwischen ihren Mitgliedern der gleiche Abstand besteht. Angenommen, man verwendet den Abstand 3, dann gehören die Elemente 1, 4, 7, ... in eine Gruppe, genauso 2, 5, 8, ... und 3, 6, 9, ... Diese Gruppen werden nun einzeln sortiert, dann wird der Abstand verringert und die Vorgehensweise wiederholt. Das macht man so lange, bis der Abstand 1 ist, so dass im letzten Schritt gar keine Unterteilung mehr stattfindet.

Es stellt sich die Frage, nach welcher Methode die Gruppen sortiert werden. Die Anwort ist Bubblesort nicht unähnlich, jedoch etwas schlecht zu erkennen. Für jeden Schritt gibt die Variable Gap den Abstand zwischen den Elementen an. Die Variable Index läuft alsdann von Gap + 1 bis zum hinteren Ende der Tabelle durch. Für jeden Wert von Index wird ein Sortierlauf gestartet, mit Index als oberer Grenze - welche Gruppe dabei sortiert wird, hängt also von Index ab. Die Variable j wird auf das nächstkleinere Element der Gruppe initialisiert (also Index - Gap), dann wird es mit dem nächsthöheren verglichen und ggf. vertauscht. Dann geht j wieder einen Schritt nach unten (Schrittweite Gap), usw., bis j < 0 ist.

Wenn Index das n. Element einer Gruppe zu fassen hat, sind also alle Elemente (dieser Gruppe) unterhalb dessen sortiert. Ist Index bis zum Ende durchgelaufen, sind alle Gruppen sortiert. Fragt sich nur noch, nach welchen Regeln der Abstand verringert werden soll. Im Beispiel unten wird er immer halbiert; es hat sich jedoch in Messungen ergeben, dass Shellsort schneller ist, wenn die Abstände immer ungerade sind (warum weiß der Himmel), wofür man ja sehr leicht sorgen kann. Die Schrittweiten wären dann z. B. 63, 31, 15, 7, 3, 1.

Einfacher zu verstehen als alle umständlichen Erklärungen ist wahrscheinlich der Pascal-Quelltext selbst: 

procedure ShlSort(var Table: TableArray);

var
   Gap   : LongInt;

   x     : LongInt;

   Index : LongInt;

   j     : LongInt;

begin
   Gap := TableSize div GapDivisor;

   while Gap > 0 do
   begin
      for Index := Gap + 1 to TableSize do
      begin
         j := Index - Gap;

         while j > 0 do
         begin
            if Table[j] <= Table[j + Gap] then
               j := 0

            else
            begin
               x := Table[j];

               Table[j] := Table[j + Gap];

               Table[j + Gap] := x;

            end;

            Dec(j,Gap);

         end;

      end;

      Gap := Gap div GapDivisor;

   end;

end;

Shellsort legt eine bemerkenswerte Geschwindigkeit an den Tag, so dass man es schon fast zu den "schnellen" Algorithmen zählen könnte. Leider kommt es mit sehr großen Arrays überhaupt nicht zurecht - heutzutage könnte man sagen, es skaliert schlecht.

Quicksort

Wenn eine Methode schon Quicksort heißt, weckt das gewisse Erwartungen, und sie werden auch nicht enttäuscht: Quicksort ist in den meisten Fällen der schnellste Algorithmus. Was steckt dahinter?
Die Idee ist recht einfach: Man greift sich ein beliebiges Element des Arrays heraus - beispielsweise das mittlere - und teilt das Array in zwei Gruppen: Eine mit den Elementen, die größer sind als das herausgegriffene, und eine mit den kleineren (oder gleichen). Diese beiden Hälften übergibt man vertrauensvoll wiederum an Quicksort. Es handelt sich also um eine rekursive Funktion. Naja, und irgendwann sind die Arrays nur noch 1 Element groß und trivialerweise sortiert.

Um nun nicht immer neue Arrays erzeugen zu müssen, arbeitet Quicksort mit linker und rechter Grenze. Dafür müssen dann natürlich alle Elemente, die größer sind als das herausgegriffene, in den unteren Teil verschoben werden, die anderen in den oberen Teil. Darin besteht die eigentliche Arbeit von Quicksort: 

procedure QSort(Links, Rechts : integer);

var
   i, j : integer;

   x, w : Byte;

begin
   i := Links;

   j := Rechts;

   x := Table[(Links + Rechts) div 2];

   repeat
      while Table[i] < x do Inc(i);

      while Table[j] > x do Dec(j);

      if i <= j then
      begin
         w := Table[i];

         Table[i] := Table[j];

         Table[j] := w;

         Inc(i);

         Dec(j);

      end;

   until i > j;

   if Links < j then QSort(Links,j);

   if Rechts > i then QSort(i,Rechts);

end;

In x wird zunächst das herausgegriffene Element gespeichert - in diesem Fall die Mitte. i und j sind Laufindizes, die zunächst links und rechts positioniert werden. Dann marschieren sie von beiden Seiten aufeinander zu, bis sie ein Element finden, das nicht auf ihre Seite gehört. Sollten sie sich zu diesem Zeitpunkt nicht schon überholt haben (in dem Fall würde es ja wieder stimmen), tauschen sie die beiden Elemente aus und machen weiter. Das wiederholt sich so lange, bis sich die beiden erreicht bzw. überholt haben.
Anschließend wird QSort rekursiv aufgerufen, einmal für den unteren Teil, falls ein solcher überhaupt mit mehr als einem Element existiert, einmal für den oberen Teil (dito). Da die Rekursion nicht in der innersten Schleife stattfindet, ist sie relativ harmlos, allerdings kann das natürlich schon etwas auf den Stack gehen. Daher wird manchmal eine iterative Implementation empfohlen, bei der man den Stack selbst verwaltet; besonders trivial ist das leider nicht.

Quicksort ist, wie eingangs gesagt, in der Regel das schnellste Suchverfahren. Im schlimmsten Fall jedoch, wenn das herausgegriffene x jedesmal das größte (oder kleinste) Element des Arrays ist, fällt Quicksort leider auf das Niveau von Bubblesort herab. In den meisten Situationen fällt das nicht ins Gewicht, aber in Echtzeitanwendungen muss man es schon beachten.

Heapsort

Heapsort ist ebenfalls ein sehr schneller Algorithmus, aber im Durchschnitt nicht ganz so schnell wie Quicksort. Dafür leistet es sich auch im schlimmsten Fall keinen solchen Durchhänger wie dieses, so dass sich Heapsort vor allem für Anwendungen empfiehlt, in denen Höchstzeiten eingehalten werden müssen.
Die Funktionsweise ist leider nicht ganz trivial (finde ich). Man hat sich vorzustellen, dass die Elemente des Arrays tatsächlich in einem binären Baum gespeichert sind (näheres zu Bäumen findet sich übrigens auf http://www.bastisoft.de/ in der Rubrik über dynamische Datenstrukturen). Aber keine Angst, es kommt jetzt nicht die für binäre Bäume übliche Pointer-Orgie, denn es gibt einen Trick, mit dem man (ausgeglichene) binäre Bäume direkt im Array darstellen kann. Dazu stellen wir uns erstmal vor, Element 1 des Arrays sei die Wurzel des Baums. Und weiter gilt: Jedes Element des Arrays, z. B. i, ist ein Knoten des Baumes, und seine beiden Nachfolger sind die Elemente 2 · i und 2 · i + 1. Die folgende Abbildung verdeutlicht, wie die Knoten im Array liegen:

[image: image6]


Einen solchen Baum nennt man einen "Heap" (engl. heap = Haufen), wenn jeder Knoten größer ist als seine Nachfolger. Die hintere Hälfte des Arrays - egal, in welchem Zustand er sich befindet - ist immer ein Heap, weil die Knoten ja gar keine Nachfolger mehr haben. Wenn der hintere Teil des Arrays (Hälfte oder mehr) ein Heap ist, kann ich diesen um ein Element nach vorne erweitern. Damit das Ergebnis immer noch ein Heap ist, muss ich das Element am neuen Knoten allerdings im Baum "absinken" lassen (engl. to sift = sieben), wobei es immer mit dem größeren seiner Nachfolgeknoten vertauscht wird - so lange, bis beide Nachfolgeknoten kleiner sind.

Dieser Vorgang wird von der Prozedur Sift erledigt; wir werden ihn noch öfter brauchen: 

procedure Sift(L, R : LongInt);

var
   i, j : LongInt;

   x    : Byte;

begin
   i := L;

   j := 2 * L;

   x := Table[i];

   if (j < R) and (Table[j + 1] > Table[j]) then Inc(j);

   while (j <= R) and (Table[j] > x) do
   begin
      Table[i] := Table[j];

      i := j;

      j := j * 2;

      if (j < R) and (Table[j + 1] > Table[j]) then Inc(j);

   end;

   Table[i] := x;

end;

Wie man sieht, bekommt Sift eine linke Grenze, dort fängt es an, und eine rechte Grenze, an der es aufhört - auch wenn das Array vielleicht noch weiterginge (es ist jetzt noch etwas unklar, wozu das gut sein soll: Geduld!). In x wird der Wert des zu "siftenden" Elements gespeichert, i ist der aktuelle Knoten und j der Nachfolgeknoten - entweder 2 · i oder, wenn größer, 2 · i + 1.
Der erste Teil von Heapsort besteht genau darin, wie beschrieben von der Hälfte ausgehend den Heap nach vorne zu erweitern, bis das ganze Array ein Heap ist. Der zweite Teil, das eigentliche Sortieren, ist nun eigentlich auch nicht mehr schwer zu verstehen: Das erste Element des Arrays ist, weil es die Wurzel eines Heaps ist, das größte Element. Es gehört also ans Ende des Arrays. Setzen wir es doch einfach dorthin, indem wir es mit dem dortigen Element austauschen. Dieses ist nun an Position 1. Wir lassen es mit Hilfe von Sift einsinken - natürlich nur bis zum vorletzten Element, das letzte wollen wir in Ruhe lassen - und haben anschließend wieder das Maximum an Position 1 (weil die Knoten der zweiten Ebene ja auch die Maxima ihrer Unterheaps sind). Dieses vertauschen wir nun mit dem Element an vorletzter Stelle, lassen die neue Nummer 1 einsinken usw. usf. Wenn wir das ganzen Array besetzt haben, sind wir fertig, und weil das Array von unten her mit den jeweils größten Elementen aufgefüllt wurde, ist er sortiert - voilà!

Der Heapsort-Algorithmus in Pascal, die beiden Teile sind durch eine Leerzeile getrennt: 

procedure HSort;

var
   L, R  : LongInt;

   Index : LongInt;

   x     : Byte;

begin
   L := (TableSize div 2) + 1;

   R := TableSize;

   while (L > 1) do
   begin
      Dec(L);

      Sift(L,R);

   end;

   for Index := R downto 1 do
   begin
      x := Table[1];

      Table[1] := Table[Index];

      Table[Index] := x;

      Sift(1,Index - 1);

   end;

end;

Nun wird auch deutlich, warum Sift auch eine rechte Grenze mitbekommt. Eine mögliche Verbesserung besteht darin, dass man den Code von Sift gleich direkt in den Heapsort-Algorithmus schreibt. Dies sollte kein großes Problem sein, nur mit den Variablennamen muss man ein bisschen aufpassen.
Ganz schön viel Aufwand für einen Sortieralgorithmus, könnte man sagen. Andererseits ist es halb so schlimm, wenn man es einmal verstanden hat. Und der Trick mit dem Unterbringen von binären Bäumen in Arrays ist ja auch etwas, das man sich merken kann.

Bucketsort

Hörte es sich bis jetzt so an, als wären Quicksort und Heapsort die schnellsten Suchverfahren, so wirkt es vielleicht verwunderlich, aber das eher selten anzutreffende Bucketsort verspricht, noch schneller zu sein. Die Sache hat - natürlich - einen Haken: Bucketsort benötigt zusätzlichen Platz. Und zwar umso mehr, je größer die Menge der möglichen Schlüssel ist. Warum das so ist, wird gleich deutlich.
Das Prinzip ist einfach: In einem zusätzlichen Array namens Bucket gibt es für jeden möglichen Schlüssel ein "Fach". Dann wird der Array Table von vorn bis hinten durchgegangen, und jedes Element wird in sein Fach gelegt. Anschließend braucht man nur noch den Inhalt der Fächer nacheinander zurück in das Array zu kopieren; dabei sollten die Fächer natürlich in der richtigen Reihenfolge sein. Anschließend ist das Array sortiert.

Leider ergibt sich eine Reihe von Problemen: Wie z. B. speichere ich mehrere Elemente in einem Feld des Bucket-Arrays? Diese Elemente müssen ja nicht unbedingt gleich sein, sie haben nur den gleichen Sortierschlüssel! Hierfür gibt es verschiedene Lösungen, aber die naheliegendste ist, eine linked List zu verwenden.

Zum anderen möchte man häufig Strings sortieren, aber selbst von Strings der Länge 10 gibt es 2610 = 1.411671 · 1014 Möglichkeiten. Von Standard-Strings der Länge 255 will ich mal lieber gar nicht anfangen... Auch das Sortieren von LongInts mit ihren 32 Bit Länge dürfte wenig spaßig sein. Ein Ausweg besteht darin, den Schlüssel in seine Einzelelemente aufzuteilen und mehrmals zu sortieren; bei Strings wird also zuerst nach dem letzten Zeichen sortiert, dann nach dem vorletzten usw. Bei Zahlen kann man entsprechend nach einzelnen Bits sortieren.

Um den Algorithmus zu verdeutlichen, will ich aber ein viel simpleres Beispiel nehmen: Es werden Bytes sortiert, der Bereich, innerhalb dessen sie Werte annehmen dürfen, ist also 0 bis 255. Eine starke Vereinfachung besteht darin, daß die Zahlen nur aus dem Schlüssel bestehen können, und zwei Elemente mit dem gleichen Schlüssel nicht unterscheidbar sind. Daher brauchen sie nicht in den Fächern gespeichert zu werden - der Index des Faches ist schon der Wert, und im Fach wird einfach gezählt, wie oft die betreffende Zahl schon aufgetreten ist. So oft wird sie dann beim Zurückschreiben in das Array wiederholt.

Und so sieht die Prozedur aus: 

procedure BuckSort;

var
   Index     : LongInt;

   BuckIndex : LongInt;

begin
   for BuckIndex := 0 to 255 do Bucket[BuckIndex] := 0;

   for Index := 1 to TableSize do
      Inc(Bucket[Table[Index]]);

   Index := 1;

   BuckIndex := 0;

   while Index <= TableSize do
   begin
      while Bucket[BuckIndex] = 0 do
         Inc(BuckIndex);

      Table[Index] := BuckIndex;

      Dec(Bucket[BuckIndex]);

      Inc(Index);

   end;

end;

In dieser Version ist Bucketsort ziemlich schnell. Eine andere Anwendung, in der es glänzt, ist z. B. das Sortieren von Strings nach Anfangsbuchstaben. Insgesamt jedoch fordert der Geschwindigkeitsvorteil seinen Tribut in Form von allerlei Umständlichkeiten, die einem mit Quicksort und Heapsort erspart bleiben. Das ist auch der Grund dafür, daß Bucketsort zumeist nur in Spezialfällen angewendet wird.

